



Clase 3 Dashboard minero, importando una carpeta

OBJETIVOS DEL TABLERO

- ¿Qué es más importante: saber cuántas toneladas se produjeron o cómo evoluciona en el tiempo?
- ¿Cómo comparo ramas de actividad tan distintas entre sí?

1. IPI Producción Litio

- Qué es:** Serie específica que mide la evolución de la producción de litio (generalmente como índice base 100 en un año de referencia).
- Interpretación:**

- 1) Permite ver la tendencia mensual o anual del litio.
- 2) Se puede graficar como línea de tiempo o barra mensual.
- 3) Es útil para responder: ¿está aumentando la producción de litio en comparación con meses anteriores? ¿Cuál es la tasa de crecimiento interanual?

2. IPI Ramas de Actividad

- Qué es:** Desagregación del IPI Minero en distintas ramas (ejemplo: litio, oro, plata, cobre, hidrocarburos no convencionales, etc.).
- Interpretación:**

- 1) Permite **comparar el litio con otras ramas de la minería**.
- 2) Sirve para analizar la **importancia relativa del litio** en la minería argentina.
- 3) También se puede detectar **qué rama tracciona el crecimiento** general del sector.

3. IPI Series Minero

1. **Qué es:** Conjunto de series históricas completas del índice (todos los meses, varias ramas).
2. **Interpretación:**

- 1) Es la **base de datos madre**, de la cual se derivan las otras tablas.
- 2) Te permite construir series de tiempo largas (varios años) para análisis de tendencia.
- 3) Sirve para armar **gráficos dinámicos** (por rama, por período).

4. IPI Variaciones Minero



1. **Qué es:** Las tasas de variación del índice (mensual, interanual, acumulado).
2. **Interpretación:**

- 1) En vez de valores absolutos del índice, muestra **crecimientos o caídas en %**.
- 2) Es clave para los **KPI del tablero**: si el litio creció un 20% interanual, es mucho más ilustrativo que mostrar solo el índice.
- 3) Permite comparar velocidad de crecimiento de distintas ramas.

5. Interpretación de las tablas y sus contenidos

1. IPI Producción Litio

- **Campos:** Período, Carbonato de litio (tn), Carbonato de litio (números índice).
- **Lectura:**
 - tn = dato **real** de producción en toneladas.
 - números índice = serie **normalizada** (generalmente base 100 en un año).
- **Uso en el dashboard:**
 - Mostrar la **producción real (tn)** en un gráfico de barras o línea.
 - Mostrar la **evolución del índice** en paralelo para compararlo con el resto de ramas.
 - KPI: toneladas del último período y su variación vs. el período anterior.

2. IPI Ramas de Actividad

- **Campos:** Fecha, Serie, Índice.
- **Lectura:**
 - Cada “Serie” es una rama de actividad minera (litio, oro, plata, etc.).
 - Índice = evolución normalizada de cada rama.
- **Uso en el dashboard:**
 - Comparar litio con otras ramas en un mismo período.
 - Graficar la contribución o evolución de cada rama.
 - Ranking: qué rama creció más/menos.

3. IPI Series Minero

- **Campos:** Fecha, Serie, Índice.
- **Lectura:**
 - Muy similar al anterior, pero con series más largas o consolidadas (la base histórica).
- **Uso en el dashboard:**
 - Línea de tiempo con selector de “Serie” (para ver litio, oro, etc.).



- Gráfico comparativo (ej: litio vs. total minería).

4. IPI Variaciones Minero

- **Campos:** Fecha, Serie, Variación.
- **Lectura:**
 - Variaciones expresadas en %, ya calculadas (no hace falta hacer el cálculo de tasa).
- **Uso en el dashboard:**
 - Tarjetas KPI: variación mensual, interanual o acumulada.
 - Barras con variaciones por rama.
 - Semáforo: verde si crece, rojo si cae.

6. Modelado Relacional → creamos la tabla Fechas

Tabla	Columnas principales	Tipo de relación sugerida
IPI_Producción_Litio	Período, Carbonato de litio tn, Carbonato de litio índice	Relacionar → período con la tabla fechas
IPI_Ramas	Fecha, Serie, Índice	Relacionar → período con la tabla fechas
IPI_Series	Fecha, Serie, Índice	Relacionar → período con la tabla fechas
IPI_Variaciones	Fecha, Serie, Variación	Relacionar → período con la tabla fechas

Creamos la tabla de fechas → Modelado → Nueva Tabla y le damos por nombre Calendario

```
Calendario =  
ADDCOLUMNS (  
    CALENDAR (DATE (2020,1,1), DATE (2025,12,31)),  
    "Año", YEAR ([Date]),  
    "Mes", MONTH ([Date]),  
    "MesNombre", FORMAT ([Date], "MMMM"),  
    "Trimestre", QUARTER ([Date])  
)
```

7. Esquema del tablero de control

1. Cabecera (KPIs) → Borde #533E1C; Cambria Math; 15, Blanco, negrita. Sombra interior en blanco

- 1) Producción litio en toneladas (último período).
- 2) Variación mensual (%) litio.
- 3) Variación interanual (%) litio.
- 4) IPI minero total (último índice).

2. Gráfico de evolución temporal



- 1) Línea de producción litio (tn).
- 2) Línea de índice litio vs. índice minería total.

3. Comparación de ramas

- 1) Barras horizontales con índice del último período por rama.
- 2) Opcional: mapa de calor con ramas en filas y meses en columnas.

4. Variaciones mineras

- 1) Tarjetas con variaciones (%).
- 2) Gráfico de columnas con variaciones por rama.

8. Fórmulas DAX

1. Producción de litio en tn

```
Producción Litio (tn) =  
SUM( IPI_Produccion_Litio[Carbonato de litio tn] )  
Índice (para cualquier serie)  
Índice =  
SUM( IPI_Ramas[Índice] )
```

2. Última producción de litio (tn)

```
Litio Toneladas Último =  
CALCULATE(  
    MAX( IPI_Produccion_Litio[Carbonato de litio tn] ),  
    FILTER( IPI_Produccion_Litio,  
        IPI_Produccion_Litio[Período] =  
        MAX(IPI_Produccion_Litio[Período]) )  
)
```

Caso contrario también servirían

```
Producción litio último mes  
Litio Último (tn) =  
CALCULATE(  
    [Producción Litio (tn)],  
    LASTDATE( IPI_Produccion_Litio[Período] )  
)  
Índice último (genérico para cualquier serie)  
Índice Último =  
CALCULATE(  
    [Índice],  
    LASTDATE( IPI_Ramas[Fecha] )  
)
```



3. Variación mensual litio (usando tn)

```
Variación Mensual Litio =  
VAR MesActual = MAX( IPI_Produccion_Litio[Carbonato de  
litio tn] )  
VAR MesAnterior =  
    CALCULATE(  
        MAX( IPI_Produccion_Litio[Carbonato de litio tn] ),  
        DATEADD( IPI_Produccion_Litio[Período], -1, MONTH )  
    )  
RETURN  
DIVIDE( MesActual - MesAnterior, MesAnterior, 0 )
```

También servirían

```
Variación mensual (%)  
Variación Mensual % =  
VAR Actual = [Índice]  
VAR Anterior = CALCULATE( [Índice],  
    DATEADD( IPI_Ramas[Fecha], -1, MONTH ) )  
RETURN  
DIVIDE( Actual - Anterior, Anterior, 0 )  
Variación interanual (%)  
Variación Interanual % =  
VAR Actual = [Índice]  
VAR AñoAnterior = CALCULATE( [Índice],  
    DATEADD( IPI_Ramas[Fecha], -1, YEAR ) )  
RETURN  
DIVIDE( Actual - AñoAnterior, AñoAnterior, 0 )
```

4. Variación interanual del litio usando tn

```
Variación Interanual Litio =  
VAR AñoActual = MAX( IPI_Produccion_Litio[Carbonato de  
litio tn] )  
VAR AñoAnterior =  
    CALCULATE(  
        MAX( IPI_Produccion_Litio[Carbonato de litio tn] ),  
        DATEADD( IPI_Produccion_Litio[Período], -1, YEAR )  
    )  
RETURN  
DIVIDE( AñoActual - AñoAnterior, AñoAnterior, 0 )
```

5. Último índice por rama

```
Índice Último =  
CALCULATE(  
    MAX( IPI_Ramas[Índice] ),  
    FILTER( IPI_Ramas, IPI_Ramas[Fecha] =  
        MAX( IPI_Ramas[Fecha] ) )  
)
```

Gráfico de barras: **Eje = Serie, Valor = Índice Último.**



6. Variación ya calculada

Si en IPI_Variaciones ya está el campo Variación, podemos usarlo directo en una tarjeta o gráfico.

Pero si queremos la **última variación de cada serie**:

```
Variación Última =  
CALCULATE(  
    MAX( IPI_Variaciones[Variación] ),  
    FILTER( IPI_Variaciones, IPI_Variaciones[Fecha] =  
        MAX(IPI_Variaciones[Fecha]) )  
)
```

7. Medidas comparativas

```
Índice último por serie (ranking ramas)  
Índice Último por Serie =  
CALCULATE(  
    [Índice],  
    FILTER(  
        IPI_Ramas,  
        IPI_Ramas[Fecha] =  
        CALCULATE( MAX( IPI_Ramas[Fecha] ),  
            ALL( IPI_Ramas ) )  
    )  
)
```

Esto sirve para un gráfico de barras: eje = Serie, valor = Índice Último por Serie.

8. Variaciones desde tabla IPI_Variaciones → usar con tarjetas

```
Variación Actual =  
CALCULATE(  
    MAX( IPI_Variaciones[Variación] ),  
    IPI_Variaciones[Fecha] =  
    CALCULATE( MAX( IPI_Variaciones[Fecha] ),  
        ALL( IPI_Variaciones ) )  
)
```

9. Armado de Tablero Decisional

- Encabezado (KPIs principales) → **Tarjetas**: muestra el estado actual

1. Producción Litio (tn último mes)

- Medida: Litio Toneladas Último

2. Variación Mensual (%) Litio

- Medida: Variación Mensual Litio
- Formato: porcentaje con 1 decimal.



3. Variación Interanual (%) Litio

- Medida: Variación Interanual Litio

4. IPI Minero Total (índice último mes)

- Tabla: IPI_Ramas
- Medida: Índice Último
- Filtro: Serie = "Total Minería"

- Evolución Temporal: muestra tendencia

Gráfico de líneas

- **Eje X:** Período (de IPI_Produccion_Litio)
- **Valores:**
 - Carbonato de litio tn (línea 1 → producción real)
 - Carbonato de litio índice (línea 2 → índice normalizado)

Opcional:

- Otra línea desde IPI_Series con Serie = "Total Minería" → para comparar litio vs minería total.
- Comparación entre ramas: comparan al litio con otras ramas

Gráfico de barras horizontales

- **Eje Y:** Serie (de IPI_Ramas)
- **Valor:** Índice Último (DAX)

Esto te muestra **el último dato disponible de cada rama**, ordenado de mayor a menor.
Podés filtrar por mes con un **slicer de fecha** arriba.

- Variaciones mineras: responden a preguntas como
 - ✓ ¿Dónde crece la minería?
 - ✓ ¿Dónde cae o decrece la minería?

Gráfico de barras verticales

- **Eje X:** Serie (de IPI_Variaciones)
- **Valor:** Variación Última (DAX)

- Semáforo / Heatmap

Matriz o Tabla condicional

- Filas: Serie
- Columnas: Fecha (meses)
- Valores: Variación (%)



Con formato **condicional con colores** (verde crecimiento, rojo caída).
Esto da un **mapa de calor de variaciones** por rama y mes.

- Filtros (Slicers)
 - Fecha (con selector de mes).
 - Serie (para filtrar ramas: litio, oro, plata, total minería).

10. Diccionario de medidas

Medida	Qué mide (explicación teórica)	Dónde usarla (visual)	
Producción Litio (tn)	Suma de las toneladas de carbonato de litio producidas en el período. Es el dato real de producción , no un índice.	Línea temporal de producción o KPI de acumulado.	
Litio Último (tn)	Producción de litio (tn) en el último período disponible. Responde a la pregunta: <i>¿cuánto se produjo el último mes?</i>	Tarjeta KPI en el encabezado.	
Índice	Valor del índice de producción para cualquier rama. Es una serie normalizada , que permite comparar ramas con bases distintas.	Gráficos de líneas y comparativos entre ramas.	
Índice Último	Índice de la serie filtrada (ej: litio, total minería, oro, etc.) en el último período.	Tarjeta KPI o línea que muestre el valor actual.	
Variación Mensual %	El porcentaje de cambio del índice respecto al mes anterior. Es una medida de corto plazo (ej: “creció 5% vs. el mes pasado”).	Tarjeta KPI con formato %.	
Variación Interanual %	El porcentaje de cambio del índice respecto al mismo mes del año anterior. Indica la tendencia a largo	Tarjeta KPI o gráfico de columnas comparativas año contra año.	



	plazo (ej: "+12% vs. julio 2023").		
Índice Último por Serie	El índice de cada rama de actividad en el último mes, en un mismo gráfico. Permite comparar ramas en un corte temporal único .	Barras horizontales (ranking ramas).	
Variación Actual	La última variación disponible de cada serie desde la tabla de variaciones. Responde a: <i>¿cómo le fue a cada rama en el último período?</i> .	Column chart o heatmap de variaciones.	
Variación Mensual Litio (tn)	El cambio porcentual de la producción real (tn) de litio entre el último mes y el mes anterior.	Tarjeta o gráfico de barras (para explicar evolución real vs índice).	

Aclaraciones sobre el trabajo

1. Diferenciar valores reales vs. índices

- *Valores reales* (tn de litio) → nos dicen cuánto se produjo.
- *Índices* → normalizan la información para poder comparar distintas ramas o con un total.

2. Diferenciar valores absolutos vs. variaciones

- *Valores absolutos* (ej: 1.200 tn de litio) → estado actual.
- *Variaciones* (ej: +5% mensual, +12% interanual) → cómo cambia respecto al pasado.

3. Diferenciar corto plazo vs. largo plazo

- *Mensual* → qué pasó de un mes a otro.
- *Interanual* → qué pasó comparado con el mismo mes del año anterior.

4. Uso del tablero

- **Encabezado** = resumen de lo más importante (último dato y variaciones).
- **Línea temporal** = tendencia (hacia dónde va).
- **Ranking ramas** = comparativo entre diferentes actividades.
- **Variaciones** = semáforo de crecimiento o caída.